
Subject: NFHS-5 Disability and health seeking
Posted by [Lincoln](#) on Tue, 08 Nov 2022 12:39:06 GMT
[View Forum Message](#) <> [Reply to Message](#)

I am trying to use the PR data set to analyse where do people with disability visit for seeking health care by type of disability.

sh73= when members of your household get sick, where do they generally go for treatment.

sh31=does any member of this household including you have any disability?

However, there are six variables for each person variable! example:

sh33a1a
sh33a1b
sh33a1c
sh33a1d
sh33a1e
sh33a1x

and there are four similar sets a, b, c, d.

Not sure how to move forward.

Please help

Regards
Lincoln

Subject: Re: NFHS-5 Disability and health seeking
Posted by [Bridgette-DHS](#) on Tue, 08 Nov 2022 13:31:59 GMT
[View Forum Message](#) <> [Reply to Message](#)

Are you asking about specific tables in the NFHS-5 final report? If so, which table?

Subject: Re: NFHS-5 Disability and health seeking
Posted by [Lincoln](#) on Tue, 08 Nov 2022 13:49:13 GMT

I am trying to create a modified table which is similar to Table-11.16

with first column as "source" and all other columns as type of disability (hearing, speech, visual , mental, mental, locomotor, others) .

However, for now if you can kindly help me recreate the "Table 2.29 Disability", it will be great.

Regards
Lincoln

Subject: Re: NFHS-5 Disability and health seeking
Posted by [Bridgette-DHS](#) on Wed, 16 Nov 2022 16:02:21 GMT
[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

Here is a Stata program to produce NFHS-5 table 2.29, on disabilities. Construction of this table is tricky because of how the disability variables in the HR file were moved into the PR file. Note that I code the outcomes 0 and 100, rather than 0 and 1, so that the means will be percentages. The tabstat command is helpful. It has a "format" option which can change the number of decimal places. I use fweight because tabstat does not allow iweight. You can get the frequencies with tab and iweight=hv005/1000000.

* Construction of NFHS-5 Table 2.29

use "C:\Users\26216\ICF\Analysis - Shared Resources\Data\DHSdata\IAPR7DFL.DTA", clear

- * Up to four people in the household may have a disability
- * The line numbers of person a, b, c, d are given by sh33a, sh33b, sh33c, sh33d
- * All the variables are repeated for all members of the household

```
gen dis_any=0
local letters a b c d e x
foreach ll of local letters {
  gen dis_`ll'=0
  replace dis_`ll'=sh33a1`ll' if hvidx==sh33a
  replace dis_`ll'=sh33b1`ll' if hvidx==sh33b
  replace dis_`ll'=sh33c1`ll' if hvidx==sh33c
  replace dis_`ll'=sh33d1`ll' if hvidx==sh33d
  replace dis_`ll'=100*dis_`ll'
```

```
replace dis_any =100 if dis_`ll'==100
}
```

```
label variable dis_any "Any"
label variable dis_a "Hearing"
label variable dis_b "Speech"
label variable dis_c "Visual"
label variable dis_d "Mental"
label variable dis_e "Locomotor"
label variable dis_x "Other"
```

```
gen age=0
replace age=1 if hv105>=5
replace age=2 if hv105>=15
replace age=3 if hv105>=25
replace age=4 if hv105>=35
replace age=5 if hv105>=50
replace age=6 if hv105>=70
replace age=7 if hv105>95
```

```
label define age 0 "0-4" 1 "5-14" 2 "15-24" 3 "25-34" 4 "35-49" 5 "50-69" 6 "70+" 7 "DK/missing"
label values age age
```

```
tabstat dis_* if hv102==1 [fweight=hv005], statistics(mean) by(hv025)
tabstat dis_* if hv102==1 [fweight=hv005], statistics(mean) by(age)
```

```
* Repeat for hv104=1, 2, 3
forvalues lsex=1/3 {
tabstat dis_* if hv102==1 & hv104==`lsex' [fweight=hv005], statistics(mean) by(hv025)
tabstat dis_* if hv102==1 & hv104==`lsex' [fweight=hv005], statistics(mean) by(age)
}
```

Subject: Re: NFHS-5 Disability and health seeking
Posted by [Lincoln](#) on Thu, 17 Nov 2022 10:53:51 GMT
[View Forum Message](#) <> [Reply to Message](#)

Many thanks for the kind support.

Regards
Lincoln
